APPENDIX D3: Other Highway Model Program Source Codes
1-MKTURN65
1-1 MKTURN65.FOR

2-MLEG
2-1 MLEG.FOR

Other Highway Model Program Source Codes
1-MKTURN65
1-1 MKTURN65.FOR

      program mkturn65

c PROGRAMMED BY KEN KALTENBACH, THE CORRADINO GROUP 11.20.2001

c-Original Version (MKTURN) by KDK (SERPM5, PB/BO 2000)

c-Modified Version (MKTURN2) by SKS for use in SERPM6

c Date: Nov 8, 2005 (MKTURN2-PRE)

c Date: Dec 20, 2005 (MKTURN2)

c

c Program to assign turn penalties on the basis of type of turn

c and facility type

c

c Type of turns are right, through and left, determined by coordinates

c Facility types are 1-digit FSUTMS (from link - to link)

c

c Read turn definitions from turndef.altin

c    cc:1 is card type

c       c is a comment

c       R defines the maximum right turn

c       L defines the minimum left turn

c       T is a tradtional TCARD

c       F is a factor for the generated turn penalties applied to

c         specified intersections.

c       blank is a turn definition

c

c    cc: 6 is from FT

c    cc:11 is to FT

c    cc:24-25 is left turn penalty in hundredths of minutes

c    cc:29-30 is through penalty in hundredths of minutes

c    cc:34-35 is right turn penalty in hundredths of minutes

c    cc:36-80 is an optional description

c

c    Penalties entered as 0 ARE NOT written.

c    An intersection MUST have more than 2 connected nodes for

c    a penalty to be applied.

c

c    F defines the MAXIMUM included angle for a Right Turn (degrees)

c    L defines the MINIMUM included angle for a Left Turn (degrees)

c

c Traditional FSUTMS "TCARDS" should be added to the turndef.altin file.

c They are read and appended to the output, and

c new penalty records based on FT and turn type are not generated for these

c intersections. PROHIBITORS should appear as TCARDS.

c

c F cards appear as intersection nodes and factors. These are useful for

c intersections with unusually high or low turn delays. All "generated"

c turn penalties for the specified intersection will be multiplied by the factor.

c  F appears in CC:1

c  The intersection node appears in CC: 2-6 (integer)

c  The factor appears in CC: 7-12 (f6.0)

c

c The new turns file is called TCARDS.altin

c

c A report is written to MKTURN.OUT

c

c MKTURN2:The program also reads few LINKS and XY attributes written from CV Network

c [Original/MKTURN Version: The original requires FSUTMS LINKS.altin and XY.altin file)

c

c altin must be entered on the command line.

c

      integer n0,kn,dir0,a0,b0,ft0,kc,zonesa,kf

      integer c(32),fc(32),dc(32)

      integer a(99999),b(99999),dir(99999),ft(99999)

      real y0,y(99999),x0,x(99999),fac(99999)

      integer tp,ttype,pen(9,9,3),tval

      integer ax(99999),bx(99999),cx(99999),px(99999)

      character*32 flinks,fxys,ftcards,fpen,fout

      character*12 fn(9)

      character*24 dtype(3)

      integer*4 narg,LUNIN

      integer*2 istat

      character*3 ya,ay

      character*12 tag

      LOGICAL EXISTS,newt

      character*80 buf

      integer r,t,l

      common RMAX,LMIN,pi

      real RMAX,LMIN,pi

c... read command line

      narg=NARGS()

      if(narg .ne. 2) then

        stop ' Put yralt on command line'

      endif

      call GETARG(1,ya,istat)

c

c     write(*,*) ' ALTIN=',ya

c

c... check for turndef file

c    if it exists, use it and run

c    otherwise quit, and the script will use

c    the existing tcards file

        fpen='mturndef.'//ya

        INQUIRE (FILE=fpen,EXIST=EXISTS)

        IF(.NOT.EXISTS) THEN

           WRITE(*,*) '*** MTURNDEF NOT PRESENT'

           WRITE(*,*) '*** MKTURN DOES NOTHING'

           WRITE(*,*) '*** SCRIPTS WILL USE TCARDS IF PRESENT'

           STOP

        ENDIF

        WRITE(*,*) '*** MTURNDEF PRESENT'

        WRITE(*,*) '*** MKTURN WILL MAKE TCARDS'

c

c... read profile.mas

C

c     write(*,*) ' Begin reading Profile.mas'

      zonesa=0

      LUNIN = 11

      OPEN (LUNIN,FILE='PROFILE.MAS',STATUS='OLD',FORM='FORMATTED')

   9  READ (LUNIN,11,END=30) TAG

   11 FORMAT (1X,A11)

      IF (TAG.EQ.'ZONESA') READ (LUNIN,*) ZONESA

   21 FORMAT (A)

      GO TO 9

   30 CLOSE (LUNIN)

      if(zonesa .eq. 0) then

          write(*,161)

161       format(' ** Fatal error, zonesa',

     &           ' missing in PROFILE.MAS')

          stop

      endif

c     write(*,*) ' Finished reading Profile.mas'

c Initialize node coordinate arrays (subscript is node number)

      do 2 i=1,99999

        x(i)=0.0

        y(i)=0.0

        fac(i)=1.0

2     continue

      kot=1 ! number of T turn cards

c make file names

c-sks      flinks='links.'//ya

c-sks      fxys  ='xy.'//ya

      flinks='M2links.DAT'

      fxys  ='M2XY.DAT'

      ftcards='tcards.'//ya

      fout='mkturn65.out'

      open(1,file=flinks)

      open(2,file=fxys)

      open(4,file=ftcards)

      open(9,file=fpen)

      open(8,file=fout)

      write(8,811) ya

811   format(' TURN PENALTY PROGRAM'/,'The Corradino Group (12.20.05)'

     &       //,'  ALTIN=',a3//)

c

c ASSIGN PENALTIES and Read Angles

      pi=4*atan(1.0)

      LMIN=1.25*pi

      RMAX=0.75*pi

      do 17 i=1,9

       do 17 j=1,9

         do 17 k=1,3

           pen(i,j,k) = 0

17    continue

19    read(9,901,end=919) buf

        write(8,901) buf

901     format(a80)

        if( (buf(1:1) .eq. 'C') .or. (buf(1:1) .eq. 'c') ) goto 19

        if( (buf(1:1) .eq. 'R') .or. (buf(1:1) .eq. 'r') ) then

          read(buf,941) RMAX

941       format(2x,f3.0)

          RMAX=pi*RMAX/180.0

          goto 19

        endif

        if( (buf(1:1) .eq. 'L') .or. (buf(1:1) .eq. 'l') ) then

          read(buf,941) LMIN

          LMIN=pi*LMIN/180.0

          goto 19

        endif

        if( (buf(1:1) .eq. 'T') .or. (buf(1:1) .eq. 't') ) then

          read(buf,602) ax(kot),bx(kot),cx(kot),px(kot)

602       format(1x,4i5)

          kot=kot+1

          goto 19

        endif

        if( (buf(1:1) .eq. 'F') .or. (buf(1:1) .eq. 'f') ) then

         read(buf,601) nf,fac(nf)

601       format(1x,i5,f6.0)

          kf=kf+1

          goto 19

        endif

        read(buf,902) i,j,l,t,r

902     format(5x,i1,4x,i1,9x,3i5)

        if(i .eq. 0) goto 19

        if(j .eq. 0) goto 19

        pen(i,j,3)=l

        pen(i,j,2)=t

        pen(i,j,1)=r

        goto 19

919   continue

      close(9)

c     write(*,836) RMAX,LMIN

      write(8,836) RMAX,LMIN

836   format(/' MAXIMUM RIGHT TURN=',f5.3,' Radians'/

     &        '  MINIMUM LEFT TURN=',f5.3,' Radians')

      write(8,*) ' ZONESA=',ZONESA

      kot=kot-1

c

c ... REPORT PENALTIES

c

      dtype(1) ='RIGHT TURNS             '

      dtype(2) ='THROUGH MOVEMENTS       '

      dtype(3) ='LEFT TURNS              '

      fn(1)='         FWY'

      fn(2)='   UNINT HWY'

      fn(3)='        None'

      fn(4)='HIGH SPD ART'

      fn(5)='      CENCON'

      fn(6)=' LOW SPD ART'

      fn(7)='        RAMP'

      fn(8)='         HOV'

      fn(9)='        TOLL'

      do 777 t=1,3

        write(8,442) dtype(t)

442     format(//'PENALTIES (hundredths of min.) BY FACILITY TYPE FOR:'

     &          /A24 /48x,'To:'/12x,9('------------'))

        write(8,443) (fn(i),i=1,9)

443     format('   From:    ',9A12/120('-'))

        do 778 i=1,9

          write(8,441) fn(i),(pen(i,j,t), j=1,9)

441       format(a12,9i12)

778     continue

777   continue

      write(8,*) kot,'T Cards'

      write(8,*) kf,'Factor Cards'

c -------------------------------------------------

c ***** read the coordinates file

      kn=0 ! highest node number

10    read(2,101,end=199) n0,x0,y0

101   format(1x,i5,2f9.0)

        x(n0)=x0

        y(n0)=y0

        if(n0 .gt. kn) kn=n0

      goto 10

199   continue

      close(2)

c     write(*,*) kn,'is the highest node number'

      write(8,*) kn,'is the highest node number'

      j=1

20    read(1,201,end=299) dir0,a0,b0,ft0

c-sks 201   format(1x,i1,2i5,6x,i1)

201   format(1x,i1,2i6,3x,i1)

c-sks9/5/06   ... trap links with no Facility Type values

      if(ft0 .eq. 0) then

        write(*,*) ' **** FATAL ERROR - No FT for this link - a b ft'

        write(*,788) a0,b0,ft0

788     format('** ',2i7,i10)

        write(8,*) ' **** FATAL ERROR - No FT for this link - a b ft'

        write(8,788) a0,b0,ft0

        stop 'MKTURN ABEND'

      endif

c -------------------------------------------------

c ***** read the link file

      a(j)=a0

      b(j)=b0

      ft(j)=ft0

      dir(j)=dir0

      j=j+1

      goto 20

299   continue

      kj=j-1

      close(1)

c     write(*,*) kj,'links'

      write(8,*) kj,'links'

c

c -------------------------------------------------

c ***** find the connections to each node and generate TCARDS ***

c     Do for every node number with a non-zero xcoord

c

      do 301 i=1,kn

        if(x(i) .ne. 0.0) then

c         ... initialize, assume max 32 connections

          kc=0 ! number of connections

          do 303 j=1,32

               c(j)=0 ! connected node

              fc(j)=0 ! facility type of connected node

              dc(j)=0 ! 0=2-way, 1=1-way intersection-node, -1=1-way node-intersection

303       continue

c         ... now find all connections, facility types and directions

          do 302 j=1,kj

            if(a(j) .eq. i) then

                  kc=kc+1  ! kc is the number of nodes connected to this intersection

               c(kc)=b(j)  ! connected node

              fc(kc)=ft(j) ! facility type of connected node

              dc(kc)=0

              if(dir(j) .eq. 1) dc(kc)=1 ! flag if 1-way

            endif

            if(b(j) .eq. i) then

                  kc=kc+1

               c(kc)=a(j)  ! connected node

              fc(kc)=ft(j) ! facility type of connected node

              dc(kc)=0

              if(dir(j) .eq. 1) dc(kc)=-1 ! flag if 1-way

            endif

302       continue

c

c         ... Now process the node to write all turns

c         "To" and "From" FTYPES must have been defined

c

          if(kc .gt. 2) then ! must have more than 2 connected nodes

          do 501 j=1,kc

            do 501 k=1,kc

               if(j .ne. k) then ! no u-turns

                 if( (dc(j) .ne. 1) .and. (dc(k) .ne. -1) ) then ! only permitted travel directions

c              ... tp is the turn direction

                   tp=ttype(x(c(j)),y(c(j)),x(i),y(i),x(c(k)),y(c(k)))

                   newt=.true.

c              ... check turns on the OLDTURNS file

                   do 460 m=1,kot

                     if( (ax(m) .eq. c(j)) .and.

     &                   (bx(m) .eq. i   ) .and.

     &                   (cx(m) .eq. c(k)) ) newt=.false.

460                continue

                   tval=pen(fc(j),fc(k),tp) ! look-up the penalty

                   tval=nint(tval*fac(i))   ! apply the factors

c              ... trap illegal turn values

                   if(tval .gt. 99999) then

                     write(*,*) ' **** FATAL ERROR - penalty too large'

                     write(*,488) c(j),i,c(k),tval

488                  format('** ',3i5,i10)

                     write(8,*) ' **** FATAL ERROR - penalty too large'

                     write(8,488) c(j),i,c(k),tval

                     stop 'MKTURN ABEND'

                   endif

c              ... don't write turns in the OLDTURNS file, and undefined penalties (0)

                   if((tval .gt. 0) .and. newt)

     &               write(4,401) c(j),i,c(k),tval! write the TCARDS

401                format('T',4i5)

                 endif

               endif

501       continue

          endif

        endif

301   continue

c

c     ... Append TCARDS TURNS TO OUTPUT

c65      do 801 j=1,kot

c65        write(4,401) ax(j),bx(j),cx(j),px(j)

c65801   continue

      do 801 j=1,kot

        write(4,402) ax(j),bx(j),cx(j),px(j)

402     format('T',4i5)

801   continue

c

      close(4)

      close(8)

      stop 'Normal MKTURN65 Completion'

      end

c------------------------------------------------------------------------------

      integer function ttype(xf,yf,xi,yi,xt,yt)

c

c Ken Kaltenbach - The Corradino Group 11.20.2001

c

c     nodes: from, through, to

c     returns the turn direction

c        - 1 is a right turn

c        - 2 is a thru movement

c        - 3 is a left turn

c

c  definitions of the permitted included angles for each type of turn

c  are passed in common in RMAX and LMIN

c  the value of pi is also passed in common

c

      real xf,yf,xi,yi,xt,yt,a1,a2,azm,ta

      common RMAX,LMIN,pi

      real RMAX,LMIN,pi

c ... get the azimuth of each leg

      a1=azm(xf,yf,xi,yi)

      a2=azm(xt,yt,xi,yi)

c      write(*,1) 'a1=',a1*180./pi,' a2=',a2*180./pi

c-sks-Kdk Compiler 1     format(a3,f0.1,a4,f0.1)

c-sks-Sks Compiler 1     format(a3,f2.1,a4,f2.1)

1      format(a3,f1.1,a4,f1.1)

c ... The difference in the azimuths is the included angle

      ta=a1-a2

      if(ta .lt. 0.0) ta=ta+2*pi   ! fix negative included angle

c ... Assign the values for right, thru and left

      ttype=1                      ! right turn

      if(ta .gt. RMAX) ttype=2  ! through

      if(ta .gt. LMIN) ttype=3  ! left turn

      return

      end

c------------------------------------------------------------------------------

      real function azm(xl,yl,xi,yi)

c

c Ken Kaltenbach - The Corradino Group 11.20.2001

c

c     azmith from intersection (i) to leg node (l)

c

c ... returns azimuth in radians

c

      real xl,yl,xi,yi,dx,dy

      integer q

      common RMAX,LMIN,pi

      real RMAX,LMIN,pi

      dy=yl-yi

      dx=xl-xi

      if(abs(dy) .ge. 0.001) then ! trap division by zero, tan=infinity at pi/2

        azm=atan(dx/dy) ! azimuth use dx/dy instead of dy/dx

      else

          if(dx .ge. 0.0) then

             azm= 0.5*pi

          else

             azm=-0.5*pi

          endif

      endif

c     corrections by quadrant

c     assume quadrant 1

      if( (dy .lt. 0) .and. (dx .ge. 0.0) ) azm=azm+pi   ! quadrant 4

      if( (dy .ge. 0) .and. (dx .lt. 0.0) ) azm=2*pi+azm ! quadrant 2

      if( (dy .lt. 0) .and. (dx .lt. 0.0) ) azm=azm+pi   ! quadrant 3

      return

      end

c------------------------------------------------------------------------------

2-MLEG

2-1 MLEG.FOR

      program mleg

c modified to find nodes that are both entrance and exit ...

      integer*4 a(999999),b(999999),c(999999)

      integer*4 ftc2(999999),LANES(999999),MERGED(999999),tlane,lanex

      integer*4 ent,exit(10),main,i,j,k,imax,hot(999999)

      character*64 fin,fout,fctl,fprn,fnod

      integer ek

      real FWYCG,FWYFU,RAMPCG,RAMPFU,RANDOM,SLOPE,scale,factor

      real FCG,FFU,RCG,RFU

      real FWYED,RAMPED 

      NAMELIST /PARAM/ FWYCG,FWYFU,RAMPCG,RAMPFU,RANDOM,SLOPE,

     &                 RAMPED,FWYED 

      FWYCG  = 4.10

      FWYFU  = 2.60

      RAMPCG = 4.10

      RAMPFU = 2.60

      RANDOM = 0.55

      SLOPE  = 0.20

      RAMPED = 0.30

      FWYED  = 0.10

      KT=0

      KR1=0

      KR2=0

      KL1=0

      KL2=0

c +++ Initialize +++      

      do 10 i=1,999999

        a(i)=0

        b(i)=0

        ftc2(i)=0

        hot(i)=0

10    continue

      if(nargs() .eq. 6) then

        CALL GETARG (1, FCTL)

        CALL GETARG (2, fin)

        CALL GETARG (3, fout)

        CALL GETARG (4, fprn)

        CALL GETARG (5, fnod)

      else

        Write(*,*) 

     & 'Must put CTL infile outfile rptfile nodefile on commandline'

        pause            

        stop

      endif

c +++ Open Files

      open(1,file=fin,status='old')

      open(2,file=fout)

      open(3,file=FCTL)

      open(5,file=fprn)

      open(6,file=fnod)

c +++ READ CONTROL

      read(3,PARAM)      

      scale  = exp(SLOPE)

c +++ write file hearder

      write(2,201) 

201   format('UNITS LENGTH = Feet'///

     &       'JUNCTIONFILE,'/

     &       '  SHOWASNEMA = 0,'/

     &       '  SHOWVOYONLY = 1,'/

     &       '  SHOWPHASECLR = 0'//)

c

C +++ Read Network Link Records      

      i=1

11    continue

C ... MERGED has a value of 0 for a normal right entrance and 1 for a left entrance

        read(1,101,end=99) a(i),b(i),ftc2(i),LANES(i),MERGED(i),hot(i)

c        write(*,*) i,a(i),b(i),ftc2(i)

        i=i+1

101     format(6i6)

      goto 11

99    imax=i-1      

      write(*,*) imax,' input links'

c +++ Find Merges      

      do 21 i=1,imax

        main=999999

         ent=999999

        exit=999999

c       check for more than 2 entrances ....

        nenters=0

        do 22 j=1,imax

c skip if the link is HOT

          ihot=hot(i) + hot(j)

c        

       if((b(i) .eq. b(j)) .and. (ihot .eq. 0)) then

       if(((ftc2(i).eq.11).or.(ftc2(i).eq.12).or.(ftc2(i).eq.91)).and.

     &          ((ftc2(j).eq.71) .or. (ftc2(j).eq.72) .or.

     &           (ftc2(j).eq.75) .or. (ftc2(j).eq.93)) ) then

              main=i

              ent=j

              node=b(i)

              nenters=nenters+1

            endif  

          endif

22      continue

c        write(5,555) nenters

555     format(' Nenters',i3)        

        if(nenters .gt. 1) then

          write(5,202) node

202      format('N=',i6,' has >2 entrances, merge/roundabout skipped')

         goto 21          

        endif            

c ... need to count the exits ...

        ek=0

        lanex=0

        do 23 k=1,imax

          if(a(k) .eq. node) then

             ek=ek+1

             lanex=lanex+LANES(k)

             exit(ek)=k

c            skip anything connected to a centroid connector by setting main=999999

             if(ftc2(k) .eq. 51) main=999999

c            skip anything connected to special transit link by setting main=999999

             if(ftc2(k) .eq. 59) main=999999

c            skip anything connected to special transit link by setting main=999999

             if(ftc2(k) .eq. 69) main=999999

          endif   

23      continue

        if(ek .gt. 2) then

          write(5,565) ek,node

565       format(' *SKIP More than 2 exits=',i2,' node=',i10)

          write(5,205) (b(exit(k)),k=1,ek)

205       format('     EXITS=',8I8)          

          goto 21

          stop

        endif

c

c if the freeway ends and there are no subsequent freeways or ramps to go to, ek=0, so skip

c    by setting main=999999

        if(ek .eq. 0) main=999999

c .. need to create a template for node with 2 or more exits !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

        if(main .ne. 999999) then

c +++ correct for lanes +++

c        (max mainline entrance or exit lanes + ramp lanes -1)

          tlane=max(LANES(main),lanex)+LANES(ent)-1

          factor=SCALE*exp(-slope*tlane)

          FCG=FWYCG*factor

          FFU=FWYFU*factor

          RCG=RAMPCG*factor

          RFU=RAMPFU*factor

c LIST OF NODES WITH INTERSECTION DATA ...          

          write(6,601) node

601       format(i8,', 1')

          KT=KT+1   

          if(merged(ent) .eq. 0) then

           if(ek .eq. 1 ) then

            write(2,502) node,a(main),a(main),FCG,FFU,FWYED,a(ent),

     &                   RCG,RFU,RAMPED,RANDOM,b(exit(1))

            KR1=KR1+1

502         format('JUNCTION,'/

     &           '  NODE = ',i6,','/

     &           '  TYPE = Roundabout,'/

     &           '  APPROACH1 = ',i6,','/

     &           '  ENABLE = 1,'/

     &           '  APPROACH = ',i6,','/

     &           '    CRITICALGAP =',F5.2,','/

     &           '    FOLLOWUPTIME =',F5.2,','/

     &           '    MOVEMENT = Through,'/

     &           '    ESTIMATEDDELAY =',F5.2,','/

     &           '  APPROACH = ',i6,','/

     &           '    CRITICALGAP =',F5.2,','/

     &          '    FOLLOWUPTIME =',F5.2,','/

     &           '    MOVEMENT = Right,'/

     &           '    ESTIMATEDDELAY =',F5.2,','/

     &            '    RANDOMNESS =',F5.2,','/

     &           '  APPROACH = ',i6,','/

     &           '    EXITONLY = 1'//)

           else

            write(2,503) node,a(main),a(main),FCG,FFU,FWYED,a(ent),

     &                   RCG,RFU,RAMPED,RANDOM,b(exit(1)),b(exit(2))

            KR2=KR2+1

503         format('JUNCTION,'/

     &           '  NODE = ',i6,','/

     &           '  TYPE = Roundabout,'/

     &           '  APPROACH1 = ',i6,','/

     &           '  ENABLE = 1,'/

     &           '  APPROACH = ',i6,','/

     &           '    CRITICALGAP =',F5.2,','/

     &           '    FOLLOWUPTIME =',F5.2,','/

     &           '    MOVEMENT = Through,'/

     &           '    ESTIMATEDDELAY =',F5.2,','/

     &           '  APPROACH = ',i6,','/

     &           '    CRITICALGAP =',F5.2,','/

     &          '    FOLLOWUPTIME =',F5.2,','/

     &           '    MOVEMENT = Right,'/

     &           '    ESTIMATEDDELAY =',F5.2,','/

     &            '    RANDOMNESS =',F5.2,','/

     &           '  APPROACH = ',i6,','/

     &           '    EXITONLY = 1,'/

     &           '  APPROACH = ',i6,','/

     &           '    EXITONLY = 1'//)

           endif

          else

           if(ek .eq. 1 ) then

            write(2,504) node,a(main),a(main),FCG,FFU,FWYED,

     &                   b(exit(1)),a(ent),RCG,RFU,RAMPED,RANDOM

            KL1=KL1+1

504         format('JUNCTION,'/

     &           '  NODE = ',i6,','/

     &           '  TYPE = Roundabout,'/

     &           '  APPROACH1 = ',i6,','/

     &           '  ENABLE = 1,'/

     &           '  APPROACH = ',i6,','/

     &           '    CRITICALGAP =',F5.2,','/

     &           '    FOLLOWUPTIME =',F5.2,','/

     &           '    MOVEMENT = Right,'/

     &           '    ESTIMATEDDELAY =',F5.2,','/

     &           '  APPROACH = ',i6,','/

     &           '    EXITONLY = 1,'/

     &           '  APPROACH = ',i6,','/

     &           '    CRITICALGAP =',F5.2,','/

     &          '    FOLLOWUPTIME =',F5.2,','/

     &           '    MOVEMENT = Left,'/

     &           '    ESTIMATEDDELAY =',F5.2,','/

     &            '    RANDOMNESS =',F5.2//)

           else

            write(2,505) node,a(main),a(main),FCG,FFU,FWYED,

     &            b(exit(1)),b(exit(2)),a(ent),RCG,RFU,RAMPED,RANDOM

            KL2=KL2+1

505         format('JUNCTION,'/

     &           '  NODE = ',i6,','/

     &           '  TYPE = Roundabout,'/

     &           '  APPROACH1 = ',i6,','/

     &           '  ENABLE = 1,'/

     &           '  APPROACH = ',i6,','/

     &           '    CRITICALGAP =',F5.2,','/

     &           '    FOLLOWUPTIME =',F5.2,','/

     &           '    MOVEMENT = Right,'/

     &           '    ESTIMATEDDELAY =',F5.2,','/

     &           '  APPROACH = ',i6,','/

     &           '    EXITONLY = 1,'/

     &           '  APPROACH = ',i6,','/

     &           '    EXITONLY = 1,'/

     &           '  APPROACH = ',i6,','/

     &           '    CRITICALGAP =',F5.2,','/

     &          '    FOLLOWUPTIME =',F5.2,','/

     &           '    MOVEMENT = Left,'/

     &           '    ESTIMATEDDELAY =',F5.2,','/

     &            '    RANDOMNESS =',F5.2//)

           endif

          endif

        endif

21    continue

c      

      write(5,1501) KT,KR1,KR2,KL1,KL2

1501  format(' TOTAL Intersection records output  = ',i5/      

     &       '         Right merge with one exit  = ',i5/

     &       '         Right merge with two exits = ',i5/

     &       '         Left  merge with one exit  = ',i5/

     &       '         Left  merge with two exits = ',i5)

      if(KT .ne. KR1+KR2+KL1+KL2) write(5,1502) KT,KR1+KR2+KL1+KL2

1502  format(' Total: ',i5,' not equal sum of components: ',i5)      

      close(1)

      close(2)

      close(5)

      close(6)

      stop

      end       
Corradino

