
HIGHWAY PATHS AND SKIMS
5-1-HPNET00A.S

;Important Note: Do Not change the Name of the PRINTO[1] and PRINTO[2] files.
;                The MKTURN2 program uses these names
; Do not change filenames or add or remove FILEI/FILEO statements using an editor. Use Cube/Application Manager.
RUN PGM=NETWORK PRNFILE="D:\fsutms\d4\S65TODMDL_PRN\CUBE\HPNET00A.PRN" MSG='Write Network Attributes for MKTURN2 & MLEG'
FILEO PRINTO[3] = "{OUTDIR}\MERGELEGS.TXT"

FILEO PRINTO[2] = "{OUTDIR}\M2LINKS.DAT"

FILEO PRINTO[1] = "{OUTDIR}\M2XY.DAT"

FILEO NETO = "{OUTDIR}\TEMFILES\S6TEM5.NET"

;Temporary AM Period transit attributes:
;     EXCLUDE=CGSTDTIME,LBTIMEAM,MBTIMEAM,EBTIMEAM,MRTIMEAM,RLTIMEAM,MVTIMEAM,SHTIMEAM
;Walk & MIDDAY Period transit attributes: WALKSPEED,WALKTIME,LBTIMEMD,MBTIMEMD,EBTIMEMD,MRTIMEMD,
;                                         RLTIMEMD,MVTIMEMD,SHTIMEMD
FILEI LINKI[1] = "{OUTDIR}\S6SPDCAP_{Year}.NET"

;Write N, X and Y coordinates for MKTURN routine
PROCESS PHASE=NODEMERGE

 _NX=NI.1.X/100

 _NY=NI.1.Y/100   

; PRINT LIST=NI.1.N(5.0), NI.1.X(7.0), NI.1.Y(7.0) CSV=F PRINTO=1
 PRINT LIST=NI.1.N(6.0), _NX(9.2), _NY(9.2) CSV=F PRINTO=1

ENDPROCESS

;Write Link Info for MKTURN Routine (Transit Only Optional Links Excluded)
PROCESS PHASE=LINKMERGE   

;  IF(LI.1.FTC2<>69)
  IF(!(LI.1.FTC2=59,69))

     _FT1=INT(FTC2/10)

    if(DIRCODE=0 & LI.1.A<LI.1.b) 

      PRINT LIST=DIRCODE(2.0),LI.1.A(6.0), LI.1.B(6.0),_FT1(4.0),FTC2(4.0),CSV=F PRINTO=2

    elseif(DIRCODE=1)

      PRINT LIST=DIRCODE(2.0),LI.1.A(6.0), LI.1.B(6.0),_FT1(4.0),FTC2(4.0),CSV=F PRINTO=2

    endif

  ENDIF

; Write out Freeqay Legs Data for Roundabout Merging that will be used by MLEG Program
if((FTC2=11,12,71,72,73,74,75,83,84,91,93,94) & (TWOWAY=0)) PRINT LIST=A(6.0),B(6.0),FTC2(6.0),NUM_LANES(6.0),LFWYMRG(6.0), PRINTO=3

; should we really include 51 ????
if(FTC2=51,59,69)  PRINT LIST=A(6.0),B(6.0),FTC2(6.0),NUM_LANES(6.0),LFWYMRG(6.0), PRINTO=3

;Delete transit-Only non-highway links for any highway modelling process
;  IF(LI.1.FTC2 = 59,69) DELETE          
;Make sure to exclude FTC2 of 69 (&59) in skimming
ENDPROCESS

ENDRUN
5-3-HPNET00B.S

; Do not change filenames or add or remove FILEI/FILEO statements using an editor. Use Cube/Application Manager.
RUN PGM=NETWORK PRNFILE="D:\fsutms\d4\S65TODMDL_PRN\CUBE\HPNET00B.PRN" MSG='Get Roundabout Node to a Link Attribute'
FILEI NODEI[2] = "{OUTDIR}\FwyMrgNode{ALT}{Year}.CSV",

VAR=N,FWYRNDNODE

FILEI LINKI[1] = "{OUTDIR}\TEMFILES\S6TEM5.NET"

FILEO NETO = "{OUTDIR}\S6HNET_{Year}.NET"

PROCESS  PHASE=INPUT, FILEI=LI.1

;Use this phase to modify data as it is read, such as recoding node numbers.
  ROUNDNODECLS=0

ENDPROCESS

PROCESS  PHASE=NODEMERGE  

; Use this phase to make computations and selections of any data on the NODEI files.
   IF(FWYRNDNODE=1) NODETYPE=5    ;Roundabout Freeway Node
ENDPROCESS

PROCESS  PHASE=LINKMERGE  

; Use this phase to make computations and selections of any data on the LINKI files.
;   IF(B.FWYRNDNODE=1) ROUNDNODECLS=1  ;Links of B-Node With a Roundabout Node
   IF((B.FWYRNDNODE=1)&(!(FTC2=83-86))) ROUNDNODECLS=1  ;Links of B-Node With a Roundabout Node
ENDPROCESS

PROCESS  PHASE=SUMMARY   

; Use this phase for combining and reporting of working variables.
ENDPROCESS

ENDRUN
5-4-HPPIL00A.S

;Do not change filenames or add or remove FILEI/FILEO statements using an editor. Use Cube/Application Manager.
;mkturn2 program creates TCARDS file from TURNDEF file
*if exist TCARDS.{Year}{ALT} del TCARDS.{Year}{ALT}

*if exist MKTURN65.log del MKTURN65.log

*if exist MKTURN65.OUT del MKTURN65.OUT

*copy {datadir}\MTURNDEF.{year}{alt}

*copy {datadir}\PROFILE.MAS

*copy {outdir}\M2LINKS.DAT

*copy {outdir}\M2XY.DAT

*{PATH1}\mkturn65 {YEAR}{ALT} > {OUTDIR}\MKTURN65.log

*if exist TCARDS.{Year}{ALT} copy /a TCARDS.{Year}{ALT} {OUTDIR}\TCARDS.{Year}{ALT}

*if exist MKTURN65.OUT copy /a MKTURN65.OUT {OUTDIR}\MKTURN65.OUT

*DEL M2LINKS.DAT

*DEL M2XY.DAT

*DEL MTURNDEF.{year}{alt}

*DEL MKTURN65.LOG

*DEL MKTURN65.OUT

*DEL TCARDS.{Year}{ALT}

*rem LIST {OUTDIR}\MKTURN65.log

;These changes are extra checks for introducing pre-assignment/distribution feedback loops.
*DEL {OUTDIR}\OPHSKIMS.{ALT}{Year}

*DEL {OUTDIR}\OPHSKIMS_SAV.{ALT}{Year}

*DEL {OUTDIR}\PKHSKIMS.{ALT}{Year}

5-5-HPMAT00A.S

; Do not change filenames or add or remove FILEI/FILEO statements using an editor. Use Cube/Application Manager.
RUN PGM=MATRIX PRNFILE="D:\fsutms\d4\S65TODMDL_PRN\CUBE\HPMAT00C.PRN" MSG='Convert FSUTMS TCARDS to Voyager Turn Penalty Format'
FILEO PRINTO[1] = "{OUTDIR}\TURNS_{Year}.PEN"

FILEI RECI = "{OUTDIR}\TCARDS.{YEAR}{ALT}",

 A=2-6, B=7-11,C=12-16, TPEN=17-21

;         fld=2-6,7-11,12-16,17-21
if (ri.tpen<>0) PRINT form=6.0,list=RI.A,RI.B,RI.C," 1",ri.TPEN/100(8.2), PRINTO=1

if (ri.tpen=0) PRINT form=6.0,list=RI.A,RI.B,RI.C," 1    -1",  PRINTO=1

ENDRUN
5-7-HPHWY00A.S

; Do not change filenames or add or remove FILEI/FILEO statements using an editor. Use Cube/Application Manager.
RUN PGM=MATRIX PRNFILE="D:\fsutms\d4\S65TODMDL_PRN\CUBE\HPMAT00B.PRN" MSG='Extract Terminal Times from ZDATA4B'
FILEO PRINTO[1] = "{OUTDIR}\EXTERNALTT.TXT"

FILEI RECI = "{OUTDIR}\RGENOUT\ZD4BS65.{alt}{Year}",

 fld[1]=1-1,fld[2]=5-8,fld[3]=94-99

;if (substr(reci,1,1)='4')
  card=substr(reci,1,1)

  taz=substr(reci,5,4)

  termtime=substr(reci,94,2)

;endif
if (card='4') print csv=t,list=taz,termtime, printo=1

ENDRUN
5-8-HPMAT00B.S
; Do not change filenames or add or remove FILEI/FILEO statements using an editor. Use Cube/Application Manager.
RUN PGM=HIGHWAY PRNFILE="D:\fsutms\d4\S65TODMDL_PRN\CUBE\HPHWY00A.PRN" MSG='Build Free Flow Skims (LOV & HOV)'
DISTRIBUTEINTRASTEP ProcessID='SERPM6ID',ProcessList=1-4

FILEI NETI = "{OUTDIR}\S6HNET_{Year}.NET"

FILEI LOOKUPI[1] = "{OUTDIR}\EXTERNALTT.TXT"

FILEO MATO[3] = "{OUTDIR}\HFSKIMS3.{ALT}{YEAR}.MAT",

  MO=21,22-25,26,30 NAME=COST,DISTANCE,TIME1,TIME2,

     SVCTIME,TOLLTIMEEQ,TERMINALTIME, dec=7*3          ;HOV 3+ Skims (Note: Cost is toll amount)
FILEO MATO[2] = "{OUTDIR}\HFSKIMS.{ALT}{YEAR}.MAT",

  MO=11,12-15,16,30 NAME=COST,DISTANCE,TIME1,TIME2,

     SVCTIME,TOLLTIMEEQ,TERMINALTIME, dec=7*3          ;HOV 2+ Skims (Note: Cost is toll amount)
FILEO MATO[1] = "{OUTDIR}\FHSKIMS.{Alt}{YEAR}.MAT",

  MO=1,2-5,6,30   NAME=COST,DISTANCE,TIME1,TIME2,

     SVCTIME,TOLLTIMEEQ,TERMINALTIME, dec=7*3          ;LOV Skims (Note: Cost is toll time amount)
FILEI TURNPENI = "{OUTDIR}\TURNS_{Year}.PEN"

ZONEMSG=100

ARRAY TERM=5,ATYPE={ZONESA}

;Assign Discrete Terminal time based on new AT (SATX)
term[1]={term1}         ; Existing CBD - SAT1
term[2]={term2}         ; High Density - SAT2 (>40)
term[3]={term3}         ; Medium Density - SAT3 (>18.5 & <=40)
term[4]={term4}         ; Low Density - SAT4 (>2.5 & <=18.5)
term[5]={term5}         ; Very Low Density - SAT5 (>=0 & <=2.5)
PROCESS PHASE=LINKREAD

  IF (A<={ZONESA}) ATYPE[A]=LI.AREA_TYPE ; ASSIGNS AN AREA TYPE TO EACH ZONE BASED ON LAST ENTRY (LIKE TTPREP)
;Period Model CTOLL Modification...
IF (('{TODMODEL}'=='YES')|('{TODMODEL}'=='Yes')|('{TODMODEL}'=='yes'))

;   CTOLLPK={CTOLL}+({DevCtollPk})               ;Peak Period CTOLL
    CTOLLOP={CTOLL}+({DevCtollOp})               ;Off-Peak Period CTOLL
  ELSE

;   CTOLLPK={CTOLL}
    CTOLLOP={CTOLL}

ENDIF

;IF (CTOLLPK<0) CTOLLPK=.01
IF (CTOLLOP<0) CTOLLOP=.01

IF (('{TODMODEL}'=='YES')|('{TODMODEL}'=='Yes')|('{TODMODEL}'=='yes'))

;    LW.RCTOLLPK=LI.RCTOLL+({DevCtollPk})         ;Peak Period CTOLL (Link Level, consider county difference)
    LW.RCTOLLOP=LI.RCTOLL+({DevCtollOp})          ;off-Peak Period CTOLL (Link Level, consider county difference)
 ELSE

;    LW.RCTOLLPK=LI.RCTOLL
    LW.RCTOLLOP=LI.RCTOLL

ENDIF

;IF (LW.RCTOLLPK<0) LW.RCTOLLPK=0.01
IF (LW.RCTOLLOP<0) LW.RCTOLLOP=0.01

;Set HOT Link Toll Rates...
; Assumption: Since no volume information does not extst on link; take average of MIN and MAX HOT Toll rate
;  IF (li.HOT=1)
;    LW.HOTTOLL=({MINHOTTOLL}+{MAXHOTTOLL})*0.5
;   ELSE
;    LW.HOTTOLL=0.0
;  ENDIF
;Set HOT Link Toll Rates...
;Based on a Final Off-Peak period HOT Lane Loadings: VCLOSE=0.75, DA=1.9%, SR2=1.84%, SR3+=96.26%
IF (li.HOT=1)

  LW.VCVAL=0.75

  LW.DA=0.0190

  LW.SR2=0.0184

  LW.SR3=0.9626

  LW.HOTTOLL_TEM={MINHOTTOLL}+({MAXHOTTOLL}-{MINHOTTOLL})/(1+EXP(6-9*LW.VCVAL)) ; LOGIT EQN
;----------Assume No Toll for 3+SR & Full Toll for SR2 & DA (note- Truck not permitted), compute weighted HOTTOLL as...
  LW.HOTTOLL=(LW.DA*1.0+LW.SR2*1.0+LW.SR3*0.0)* LW.HOTTOLL_TEM

 ELSE

  LW.HOTTOLL=0

ENDIF

;use time for impedance unless toll present, then convert toll + service time to time equivalent
  if (li.FTC2=95)                                                            ;For Toll Plaza Links
;     lw.linkcostpk=(li.cartoll*LW.RCTOLLPK*60)+li.svcminutes+(li.svcseconds/60)       ;Peak Period Linkcost
     lw.linkcostop=(li.cartoll*LW.RCTOLLOP*60)+li.svcminutes+(li.svcseconds/60)        ;Off-Peak Period Linkcost
     lw.svctime=li.svcminutes+(li.svcseconds/60)                              

     lw.MCARTOLL=LI.CARTOLL

;     lw.tolleqtimepk=(lw.MCARTOLL*LW.RCTOLLPK*60)                                     ;Peak Toll Cost Equi Time
     lw.tolleqtimeop=(lw.MCARTOLL*LW.RCTOLLOP*60)                                      ;Off-Peak Toll Cost Equi Time
    elseif(li.HOT=1)                                                        ;For HOT Links
;     lw.linkcostpk=li.time+LI.DISTANCE*LW.HOTTOLL*60*LW.RCTOLLPK                      ;Peak 
     lw.linkcostop=li.time+LI.DISTANCE*LW.HOTTOLL*60*LW.RCTOLLOP                       ;Off-peak
     LW.MCARTOLL=LI.DISTANCE*LW.HOTTOLL

;     lw.tolleqtimepk=(lw.MCARTOLL*LW.RCTOLLPK*60)                                     ;Peak Toll Cost Equi Time
     lw.tolleqtimeop=(lw.MCARTOLL*LW.RCTOLLOP*60)                                      ;Off-Peak Toll Cost Equi Time
    else                                                                    ;Non-TollPlaza & Non-HOT (Regular) Links
;     lw.linkcostpk=li.time
     lw.linkcostop=li.time

     lw.svctime=0

     lw.MCARTOLL=0                                                             

;     lw.tolleqtimepk=0
     lw.tolleqtimeop=0

  endif

;Set Exclude Group for PATH/Skim...
  if (li.FTC2=81,83-86) addtogroup=1     ;HOV 2+ persons Facilities
  if (li.FTC2=82) addtogroup=3           ;HOV 3+ persons Facilities
  if (li.FTC2=59,69) addtogroup=2        ;Transit non-highway optional links  
ENDPROCESS

PROCESS PHASE=ILOOP

; first, skim the network based on travel time, LOV
      PATH=LW.LINKCOSTOP,peni=1,excludegroup=1-3,

      MW[01]=PATHTRACE(LW.MCARTOLL),NOACCESS=999.99,dec=2,

      MW[02]=PATHTRACE(LI.DISTANCE),NOACCESS=999.99,dec=2,

      MW[03]=PATHTRACE(LI.TIME,1),NOACCESS=999.99,dec=3,         ;TIME Skim includes penalty set 1
      MW[05]=PATHTRACE(LW.SVCTIME),NOACCESS=999.99,dec=2,

      MW[06]=PATHTRACE(LW.tolleqtimeop),NOACCESS=999.99,dec=2

; next, skim the network based on travel time, HOV 2+ Persons
      PATH=LW.LINKCOSTOP,peni=1,excludegroup=2-3,

      MW[11]=PATHTRACE(LW.MCARTOLL),NOACCESS=999.99,dec=2,

      MW[12]=PATHTRACE(LI.DISTANCE),NOACCESS=999.99,dec=2,

      mw[13]=PATHTRACE(LI.TIME,1),NOACCESS=999.99,dec=3,          ;TIME Skim includes penalty set 1
      MW[15]=PATHTRACE(LW.SVCTIME),NOACCESS=999.99,dec=2,

      MW[16]=PATHTRACE(LW.tolleqtimeop),NOACCESS=999.99,dec=2

; next, skim the network based on travel time, HOV 3+ Persons
      PATH=LW.LINKCOSTOP,peni=1,excludegroup=2,

      MW[21]=PATHTRACE(LW.MCARTOLL),NOACCESS=999.99,dec=2,

      MW[22]=PATHTRACE(LI.DISTANCE),NOACCESS=999.99,dec=2,

      mw[23]=PATHTRACE(LI.TIME,1),NOACCESS=999.99,dec=3,          ;TIME Skim includes penalty set 1
      MW[25]=PATHTRACE(LW.SVCTIME),NOACCESS=999.99,dec=2,

      MW[26]=PATHTRACE(LW.tolleqtimeop),NOACCESS=999.99,dec=2

; next, convert the toll skims to travel time equivalent 7 Add Service times
;      MW[07]=(MW[01]*LW.RCTOLLOP*60)+MW[05]                      ;LOV
;      MW[17]=(MW[11]*LW.RCTOLLOP*60)+MW[15]                      ;HOV 2+ Persons Group
;      MW[27]=(MW[21]*LW.RCTOLLOP*60)+MW[25]                      ;HOV 3+ Persons Group
      MW[07]=MW[06]+MW[05]                      ;LOV
      MW[17]=MW[16]+MW[15]                      ;HOV 2+ Persons Group
      MW[27]=MW[26]+MW[25]                      ;HOV 3+ Persons Group
; next, create a terminal time matrix
  lookup name=extt, lookup[1]=1, result=2, interpolate=n,fail[1]=0,fail[2]=0, lookupi=1

  JLOOP

      MW[04]=MW[3] MW[14]=MW[13] MW[24]=MW[23]     ;set TIME2 same as TIME1
      if (i<={zonesi}&j<={zonesi})                 ;internal to internal
        mw[30]=(term[atype[i]]+term[atype[j]])

      endif

      if (i>{zonesi}&j<={zonesi})                  ;external to internal
        ttime=extt(1,i)

        mw[30]=(ttime+term[atype[j]])

      endif

      if (i<={zonesi}&j>{zonesi})                  ;internal to external
        ttime=extt(1,j)

        mw[30]=(term[atype[i]]+ttime)

      endif

      if (i>{zonesi}&j>{zonesi})                   ;external to external
        ttimei=extt(1,i)

        ttimej=extt(1,j)

        mw[30]=(ttimei+ttimej)

      endif

  ENDJLOOP

;==Compute Intrazonals Values - Average of the TWO Nearest Zones
;LOV Skims:
  MW[01][i] = 0                    ;Intrazonal COST/CARTOLL - LOV
  MW[02][i] = lowest(2,2) * 0.25   ;Intrazonal DISTANCE - LOV
  MW[03][i] = lowest(3,2) * 0.25   ;Intrazonal TIME1 (w/o terminal) - LOV
  MW[04][i] = lowest(4,2) * 0.25   ;Intrazonal TIME2 (w/o terminal) - LOV
  MW[05][i] = lowest(5,2) * 0.25   ;Intrazonal SVCTIME - LOV
  MW[06][i] = lowest(6,2) * 0.25   ;Intrazonal TOLLTIMEEQ - LOV
  MW[07][i] = lowest(7,2) * 0.25   ;Intrazonal TOLLTIMEEQ+SVCTIME - LOV
;HOV 2+ Persons Skims:
  MW[11][i] = 0                    ;Intrazonal COST/CARTOLL - HOV 2 persons Grp
  MW[12][i] = lowest(12,2) * 0.25  ;Intrazonal DISTANCE - HOV 2 persons Grp
  MW[13][i] = lowest(13,2) * 0.25  ;Intrazonal TIME1 (w/o terminal) - HOV 2 persons Grp
  MW[14][i] = lowest(14,2) * 0.25  ;Intrazonal TIME2 (w/o terminal) - HOV 2 persons Grp
  MW[15][i] = lowest(15,2) * 0.25  ;Intrazonal SVCTIME - HOV 2 persons Grp
  MW[16][i] = lowest(16,2) * 0.25  ;Intrazonal TOLLTIMEEQ - HOV 2 persons Grp
  MW[17][i] = lowest(17,2) * 0.25  ;Intrazonal TOLLTIMEEQ+SVCTIME - HOV 2 persons Grp
;HOV 3+ Persons Skims:
  MW[21][i] = 0                    ;Intrazonal COST/CARTOLL - HOV 3 persons Grp
  MW[22][i] = lowest(22,2) * 0.25  ;Intrazonal DISTANCE - HOV 3 persons Grp
  MW[23][i] = lowest(23,2) * 0.25  ;Intrazonal TIME1 (w/o terminal) - HOV 3 persons Grp
  MW[24][i] = lowest(24,2) * 0.25  ;Intrazonal TIME2 (w/o terminal) - HOV 3 persons Grp
  MW[25][i] = lowest(25,2) * 0.25  ;Intrazonal SVCTIME - HOV 3 persons Grp
  MW[26][i] = lowest(25,2) * 0.25  ;Intrazonal TOLLTIMEEQ - HOV 3 persons Grp
  MW[27][i] = lowest(27,2) * 0.25  ;Intrazonal TOLLTIMEEQ+SVCTIME - HOV 3 persons Grp
;  LIST = "AREA TYPE FOR ZONE",I(5.0)," IS ",ATYPE[I](5.0)
ENDPHASE

ENDRUN
5-9-HPMAT00E.S
; Do not change filenames or add or remove FILEI/FILEO statements using an editor. Use Cube/Application Manager.
RUN PGM=MATRIX MSG='Revised GM Skim for TPK, I95 and I75 externals'
FILEI MATI[1] = "{OUTDIR}\TEMFILES\FHSKIMTT.TEM"

FILEO MATO[1] = "{OUTDIR}\FHSKIMTT.{ALT}{Year}",

mo=11-15,NAME=COST,DISTANCE,TIME,TIME2,TIME2WTT,FORMAT=TRANPLAN

; The MATRIX module does not have any explicit phases.  The module does run within an implied ILOOP
; where I is the origin zones.  All user statements in the module are processed once for each origin.
; Matrix computation (MW[#]=) are solved for all values of J for each I.  Thus for a given origin zone I
; the values for all destination zones J are automatically computed.  The user can control the computations
; at each J by using a JLOOP.
;Run51:
;a=0.075 ; slope
;b=4.0   ; bias
;c=0.5   ; minfactor
;d=1.0   ; maxfactor
;Run52-60:
;a=0.085 ; slope
;b=3.5   ; bias
;c=0.45   ; minfactor
;d=1.0   ; maxfactor
;Run61:
;a=0.075 ; slope
;b=60   ; bias
;c=0.45   ; minfactor
;d=1.2   ; maxfactor
;Run62:
a=0.075 ; slope
b=60   ; bias
c=0.40   ; minfactor
d=1.2   ; maxfactor
MW[11]=mi.1.1   ;Cost
MW[12]=mi.1.2   ;Distance
MW[13]=mi.1.3   ;Time
;Note TIME2 (4th Table/Skim is used in Autocon)
MW[14]=mi.1.4   ;TIME2
;Adjust TIME2TT (Used in G/M) Skim for TPK,I95 & I75 Externals...
mw[1]=mi.1.5              ;travel time in minutes times 100
mw[5]=mi.1.5 * 0.01

jloop

  if(mw[1]<99999)

    if((j = 4203) |(j = 4204) | (j = 4239) | (i = 4203) |(i = 4204) | (i = 4239))

;Run51-60  Logit Equn:     mw[3]=d+(c-d)/(1+exp(b-a*mw[5]))
      mw[3]=d+(c-d)/(1+exp(-a*(mw[5]-b)))

      mw[15]=mw[1]*mw[3]

    else

      mw[15]=mw[1]

      mw[3]=1.0

    endif

  else

      mw[15]=mw[1]

      mw[3]=1.0

  endif

endjloop

ENDRUN
5-10-HPPIL00B.S

; Do not change filenames or add or remove FILEI/FILEO statements using an editor. Use Cube/Application Manager.
*echo Highway Paths and Skims Done                                   = %date%  %time% >>{OUTDIR}\runtime.prn

Corradino

